
Change  1
	Function name
	f_TC_12_2a_IMS1   

	Reason for change
	We need to let the IP PTC that the unprotected port is not to be closed  

	Summary of change
	Feed in new template parameter to indicate such extra handling  

	TTCN module
	IMS/12/IMS_CC_CallControlTestcases.ttcn 

	MCC160 Comment
	


Before change
  function f_TC_12_2a_IMS1() runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, IMS_Security);  // prose specifies only Annex C.2 => not applicable for GIBA

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_Ims_3gpp_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_CC_AbortCall();    // to abort call establishment procedure

    f_IMS_GlobalRelease();   // @sic R5s141339 change 2 - additional change sic@ */

    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }      
After change
  function f_TC_12_2a_IMS1() runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED, IMS_Security);  // prose specifies only Annex C.2 => not applicable for GIBA

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_Ims_3gpp_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_CC_AbortCall();    // to abort call establishment procedure

    f_IMS_GlobalRelease(cas_IMS_SecRegInfoRelease_REQ);   // @sic R5s141339 change 2 - additional change sic@ */

    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }     

Change 2
	Function name
	n/a   

	Reason for change
	Extra template for sending must be defined

	Summary of change
	Added new template to module IMS_IP_Config  

	TTCN module
	Common/IMS/IMS_IP_Config.ttcn 

	MCC160 Comment
	


Before change
    template (value) IMS_CONFIG_REQ cas_IMS_CloseTCP_REQ := {CloseTCP := true};                                  /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */
  template (value) IMS_CONFIG_REQ cas_IMS_RegInfoRelease_REQ := {RegInfoRelease := true};                      /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */  

After change
    template (value) IMS_CONFIG_REQ cas_IMS_CloseTCP_REQ := {CloseTCP := true};                                  /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

  template (value) IMS_CONFIG_REQ cas_IMS_RegInfoRelease_REQ := {RegInfoRelease := true};                      /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS) */

  template (value) IMS_CONFIG_REQ cas_IMS_SecRegInfoRelease_REQ := {SecRegInfoRelease := true};              
Change 3 

	Function name
	n/a   

	Reason for change
	Need to add new alternative to union type IMS_CONFIG_REQ  to indicate new release type

	Summary of change
	Added SecRegInfoRelease  

	TTCN module
	Common/IMS/IMS_ASP_TypeDefs.ttcn 

	MCC160 Comment
	


Before change
  type union IMS_CONFIG_REQ {

    IPsec_SecurityKeys_Type             InstallKey,

    IMS_PortsAndSecurityConfigReq_Type  PortsAndSecurityConfig,

    Null_Type                           SecurityRelease,

    Null_Type                           RegInfoRelease,                 /* @sic R5s140123 change 9 sic@ */

    Null_Type                           CloseTCP                        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

  };

After change
  type union IMS_CONFIG_REQ {

    IPsec_SecurityKeys_Type             InstallKey,

    IMS_PortsAndSecurityConfigReq_Type  PortsAndSecurityConfig,

    Null_Type                           SecurityRelease,

    Null_Type                           RegInfoRelease,                 /* @sic R5s140123 change 9 sic@ */

    Null_Type                           CloseTCP,                        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

    Null_Type                           SecRegInfoRelease  





  };
Change 4
	Function name
	f_IMS_GlobalRelease   

	Reason for change
	Forward new release type  

	Summary of change
	Added a template parameter to definition of f_IMS_GlobalRelease such that it is not necessarily the default cas_IMS_RegInfoRelease_REQ that is forwarded as type  

	TTCN module
	Common/IMS/IMS_Procedures_Common.ttcn 

	MCC160 Comment
	


Before change
  function f_IMS_GlobalRelease() runs on IMS_PTC

  {

    f_IMS_RegisterInfo_Release();  /* release registration information at IPCAN */

    f_IMS_PTC_Reset();             /* reset ImsInfo and release security

                                      @sic R5s150031: common function f_IMS_PTC_Reset sic@ */

  }      

After change
  function f_IMS_GlobalRelease(template (value) IMS_CONFIG_REQ p_RegInfoRelease  := cas_IMS_RegInfoRelease_REQ) runs on IMS_PTC

  {

    f_IMS_RegisterInfo_Release(p_RegInfoRelease); /* release registration information at IPCAN */

    f_IMS_PTC_Reset();             /* reset ImsInfo and release security

                                      @sic R5s150031: common function f_IMS_PTC_Reset sic@ */

  }     

Change 5
	Function name
	f_IMS_RegisterInfo_Release   

	Reason for change
	Need to forward new release type  

	Summary of change
	Added new function argument, and forwarded it  

	TTCN module
	Common/IMS/IMS_IP_Config.ttcn 

	MCC160 Comment
	


Before change
  function f_IMS_RegisterInfo_Release() runs on IMS_PTC

  {

    fl_IMS_ConfigureIP(cas_IMS_RegInfoRelease_REQ,
                       car_IMS_RegInfoRelease_CNF);  /* @sic R5s140780 - MCC160 Implementation sic@ */

  }      

After change
  function f_IMS_RegisterInfo_Release(template (value) IMS_CONFIG_REQ p_RegInfoRelease  := cas_IMS_RegInfoRelease_REQ) runs on IMS_PTC

  {

    fl_IMS_ConfigureIP(p_RegInfoRelease,
                       car_IMS_RegInfoRelease_CNF);  /* @sic R5s140780 - MCC160 Implementation sic@ */

  }     

Change 6
	Function name
	n/a   

	Reason for change
	Extra template for receiving is needed

	Summary of change
	Added new template to module IP_PTC_IMS_Templates  

	TTCN module
	Common/IP_PTC/IP_PTC_IMS_Templates.ttcn 

	MCC160 Comment
	


Before change
  template (present) IMS_CONFIG_REQ car_IMS_InstallKey_REQ := {InstallKey := ?};                                        

  template (present) IMS_CONFIG_REQ car_IMS_PortsAndSecurityConfig_REQ := {PortsAndSecurityConfig := ?};                

  template (present) IMS_CONFIG_REQ car_IMS_SecurityRelease_REQ := {SecurityRelease := ?};                              

  template (present) IMS_CONFIG_REQ car_IMS_CloseTCP_REQ := {CloseTCP := true};                                         

  template (present) IMS_CONFIG_REQ car_IMS_RegInfoRelease_REQ := {RegInfoRelease := true};                                   

After change
  template (present) IMS_CONFIG_REQ car_IMS_InstallKey_REQ := {InstallKey := ?};                                          
  template (present) IMS_CONFIG_REQ car_IMS_PortsAndSecurityConfig_REQ := {PortsAndSecurityConfig := ?};                  
  template (present) IMS_CONFIG_REQ car_IMS_SecurityRelease_REQ := {SecurityRelease := ?};                                
  template (present) IMS_CONFIG_REQ car_IMS_CloseTCP_REQ := {CloseTCP := true};                                           
  template (present) IMS_CONFIG_REQ car_IMS_RegInfoRelease_REQ := {RegInfoRelease := true};                               
  template (present) IMS_CONFIG_REQ car_IMS_SecRegInfoRelease_REQ := {SecRegInfoRelease := true}; 
Change 7
	Function name
	a_IP_IMS_Config   

	Reason for change
	When IP PTC receives the message for such new handling, it should behave as intended.  

	Summary of change
	Added new clause such that TCP connection on unprotected port is not closed  

	TTCN module
	Common/IP_PTC/IP_PTC_IMS_Handler.ttcn 

	MCC160 Comment
	


Before change
   */
  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue(v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */

          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);

          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, oldestOnly, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context; R5s160735: oldestOnly sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ              /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, allProtected, v_WaitForCNF);    /* @sic R5s141154, R5s151044; R5s160735: allProtected sic@ */

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, v_WaitForCNF);   /* @sic R5s151044 sic@ */

        p_ImsServer.RegistrationAddress := omit;  /* @sic R5s160825: p_ImsServer.RegistrationAddress needs to be reset as in case of IPv6 the UE may come up with a different interface ID at the registration sic@ */

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }

    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }      
After change
  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,

                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC

  { /* @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs and protected ports are fully controlled by the IMS PTC sic@ */

    /* @sic R5s130900 - MCC160 implementation for re-authentication sic@ */

    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;

    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;

    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;

    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ

      {

        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;

        p_Port.send(cas_IMS_InstallKey_CNF);

      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;

        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;

        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue(v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */

          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);

          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);

        }

        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */

        } else {

          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */

        }

        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */

          if (not isvalue(p_ImsServer.SecurityKeys)) {

            FatalError(__FILE__, __LINE__, "security keys missing");

          }

          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */

        }

        else {

          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);

        }

        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);

      }

    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, oldestOnly, v_WaitForCNF);      /* @sic R5-145732: support of 2nd security context; R5s160735: oldestOnly sic@ */

        p_Port.send(cas_IMS_SecurityRelease_CNF);

      }

    [] p_Port.receive(car_IMS_RegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ              /* @sic R5s140123 change 8 sic@ */

      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, allProtected, v_WaitForCNF);    /* @sic R5s141154, R5s151044; R5s160735: allProtected sic@ */

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, v_WaitForCNF);   /* @sic R5s151044 sic@ */

        p_ImsServer.RegistrationAddress := omit;  /* @sic R5s160825: p_ImsServer.RegistrationAddress needs to be reset as in case of IPv6 the UE may come up with a different interface ID at the registration sic@ */

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }
       //new clause, same as preceding clause except that TCP connection on unproteced port is not closed and new template is used
       [] p_Port.receive(car_IMS_SecRegInfoRelease_REQ) -> value v_IMS_CONFIG_REQ
      {

        v_WaitForCNF := false;

        fl_IMS_Server_RemoveSecurityContext(p_ImsServer, allProtected, v_WaitForCNF);

        p_ImsServer.RegistrationAddress := omit;  

        p_Port.send(cas_IMS_RegInfoRelease_CNF);

      }
    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ        /* @sic R5s130988 change 1 - MCC160 implementation sic@ */

      {

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);

        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);

        p_Port.send(cas_IMS_CloseTCP_CNF);

      }

  }     
